VBAにおいて「Set オブジェクト変数 = Nothing」は必要か

# 1. 結論
最初に結論を述べると以下です。
- 基本的には不要だが、必要な場合もある。
- 必要/不要を都度判断して記述しても良いが、都度判断するのも思考リソースを消費するため、以下のようにルール化して記述すると良いと考える。
  1. オブジェクト変数を使用しなくなったタイミングでNothingを代入する。
  1. 循環参照がある場合は、それを解消してから上記1のルールでNothingを代入する。
  1. 外部参照のオブジェクトやWorkbookを使用した場合は、それ自身が持つクローズメソッドによりオブジェクトを破棄してから、上記1のルールでNothingを代入する。

以降、調査・検証した内容含め詳細について述べていきます。


# 2. そもそもオブジェクト変数やNothingとは
### 2-1. オブジェクト変数
オブジェクト変数には、オブジェクトへの参照、つまりオブジェクトのメモリアドレスが代入されています。[^メモリアドレス]　[^メモリ領域]
オブジェクト自体が変数に代入されているわけではありません。
そのため、オブジェクト変数は参照型に分類されます。

### 2-2. Nothing
Nothingとは、オブジェクト変数にオブジェクトへの参照が代入されていない、つまり何も参照していない状態です。
正確に言うと、オブジェクトのメモリアドレスを返す関数である、ObjPtr関数の戻り値が「0」の状態がNothingです。
![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/768789/ee7a4713-efe5-4277-45cf-ef65e8492e55.png)


# 3. 「Set オブジェクト変数=Nothing」はどういった意味か
**Set オブジェクト変数=Nothing**は、そのオブジェクト変数からオブジェクトへの参照を無くす、という意味です。
オブジェクトそのものを破棄(メモリ解放)するという意味ではありません。
オブジェクトへの参照が無くなり、参照カウント(後述します)が0になることで、結果としてそのオブジェクトは破棄されメモリ解放されます。


# 4. VBAのメモリ管理について
VBAではメモリを**参照カウント方式**で管理しています。
これはオブジェクトへの参照が追加されるたびにカウントが増加し、参照が解除されるたびにカウントが減少する、というものです。

### 4-1. 参照カウント
1つのオブジェクトが何か所から参照されているかの数を**参照カウント**といいます。
参照カウントについては以下です。

1. 参照カウントが0になった(=オブジェクトがどこからも参照されなくなった)時点で、そのオブジェクトは破棄されメモリ解放されます。
また、参照カウントが残っていたとしても、プロシージャが終了するタイミングでオブジェクト変数からオブジェクトへの参照は無くなります。
1. Endステートメントを呼び出した場合は、参照カウントが残っていたとしても、**全て**のオブジェクト変数からオブジェクトへの参照は無くなります。[^Endステートメント]
1. 参照カウント方式は他のガーベジコレクションとは違い、循環参照がある場合に参照カウントが0にならないため、オブジェクトは破棄されずメモリ解放されません。[^GC]


### 4-2. オブジェクトがクラスのインスタンスの場合について
オブジェクトがクラスのインスタンスの場合、**デストラクタ**を用意して実行させる事が出来ます。

デストラクタとは、クラスのインスタンスが破棄される時に実行されるメソッドです。
VBAでは、デストラクタをイベントプロシージャ**Class_Terminate**で定義します。
**Class_Terminate**により、クラスのインスタンスが破棄されるタイミングで実行したい処理を行わせることが出来ます。

4-1の1～3について、オブジェクトがクラスのインスタンスで、デストラクタを用意した場合は、それぞれ以下のように挙動が異なります。

1. オブジェクトが破棄される前に、クラスのインスタンスに対してデストラクタが実行されます。
1. Endステートメントはプログラムそのものを強制終了させるため、デストラクタは実行されません。
1. 循環参照がある場合、オブジェクトが破棄されないため、デストラクタは実行されません。


### 4-3. プロシージャが終了するタイミングでオブジェクト変数からオブジェクトへの参照が無くなる順番について
プロシージャが終了するタイミングでオブジェクト変数からオブジェクトへの参照が無くなる順番は、オブジェクト変数にオブジェクトへの参照が代入された順番となります。
デストラクタを用意して確認してみます。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/768789/38c22548-01cc-679f-fd94-02f2baed2a81.png)
![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/768789/2a62bdd6-4e20-3ce9-af7c-0761b5213e3c.png)

クラスのインスタンスが生成された順番で解放されました。


# 5. オブジェクト変数へNothingを代入する必要がある場合について
オブジェクト変数へNothingを代入する必要がある場合としては以下です。
- プロシージャ外で宣言したオブジェクト変数やPublic/Staticで宣言したオブジェクト変数からオブジェクトへの参照を無くしたい時

:::note warn
もし上記の他に必要な場合があるようでしたら、ご存知の方はコメント等で教えてください。
:::

また、必要がある場合ではありませんが、**そこでオブジェクトへの参照が無くなる**ということを明示的に記しておきたい場合に記述しても良いです。

### 5-1. プロシージャ外で宣言したオブジェクト変数やPublic/Staticで宣言したオブジェクト変数からオブジェクトへの参照を無くしたい時について
これらのオブジェクト変数は他のローカルのオブジェクト変数とは違い、プロシージャが終了するタイミングでオブジェクトへの参照は無くなりません。
そのため、これらのオブジェクト変数からオブジェクトへの参照を無くしたい場合は、そのオブジェクト変数へNothingを代入する必要があります。

実際に確認してみます。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/768789/d4eb54cf-d986-0363-0e62-007407049dc2.png)
![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/768789/cc165654-6be7-deb3-4b6a-cd4a95a61ec3.png)
1回目と2回目の実行でメモリアドレスが変わっておらず、2回目の実行時にオブジェクト変数へNothingを代入する前まではオブジェクトへの参照が無くなっていませんでしたが、オブジェクト変数へNothingを代入したことで、オブジェクトへの参照が無くなりました。

Endステートメントを呼び出した場合でも確認してみます。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/768789/c8c54a56-df88-97e6-ab71-1bb8ed219fb7.png)
![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/768789/9b4fb135-bcc1-30f5-a55f-b6c475aaa8b3.png)
![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/768789/98e629f8-e039-90c3-f340-948dc5bb1a50.png)
Endステートメントを呼び出した場合でも同様に、オブジェクトへの参照が無くなりました。


# 6. 基本的にオブジェクト変数へNothingは代入しなくて良いか
5-1のケースを除けば、プロシージャが終了するタイミングでオブジェクト変数からオブジェクトへの参照は無くなるため、基本的にオブジェクト変数へNothingは代入せずとも問題ありません。

ですが、オブジェクト変数を使用しなくなったタイミングでNothingを代入することを推奨します。[^再代入]

理由としては、使用しなくなったオブジェクト変数をそのままにしておくと、オブジェクト変数はそのオブジェクトを参照し続けており、その分メモリが専有され、効率が落ちる可能性があるためです。
そのため、オブジェクト変数を使用しなくなったタイミングでNothingを代入し、早期段階でメモリ解放することで、効率化に繋げると良いです。
また、オブジェクト変数を使用しなくなったタイミングでNothingを代入することで、それ以降の処理でそのオブジェクト変数を使用しないことを明示することにもなるため、保守性の向上にも繋がります。


# 7. オブジェクト変数へNothingを代入してはいけない場合はあるか
オブジェクト変数へNothingを代入してはいけない場合としては以下です。

- 循環参照がある場合に、それを解消しないままNothingを代入する
- 外部参照のオブジェクトやWorkbookがある場合に、それを破棄しないままNothingを代入する

:::note warn
もし上記の他に記述してはいけない場合があるようでしたら、ご存知の方はコメント等で教えてください。
:::

### 7-1. 循環参照がある場合について
循環参照がある場合、その対象のオブジェクト変数にNothingを代入しただけでは、対象のオブジェクトは破棄されずメモリ解放されません。
```
Sub loopTest1()
    Dim i As Long
    Dim c As Collection
    Dim d As Collection
    
    For i = 1 To 100000
        Set c = New Collection
        Set d = New Collection
        
        c.Add d
        d.Add c
        '↑循環参照
        
        Debug.Print ObjPtr(c)
        Debug.Print ObjPtr(d)
        '↑それぞれのメモリアドレスが返る
        
        Set d = Nothing
        Set c = Nothing
        '↑Nothingを代入しても対象のオブジェクトは破棄されずメモリ解放されない
        
        Debug.Print ObjPtr(c) '0
        Debug.Print ObjPtr(d) '0
    Next
End Sub
```
上記コードの実行前後でメモリ使用量を比較してみると、実行後はメモリ使用量がかなり増加していることが確認できます。
- 実行前
![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/768789/c9fee80c-33cf-508d-1ea1-282b0774419e.png)
- 実行後
![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/768789/c5c11327-755d-faf3-339c-e25f370412ba.png)

またこの場合、プロシージャが終了するタイミングやEndステートメントを呼び出した場合でも、対象のオブジェクトは破棄されずメモリ解放されません。
```
Sub loopTest2()
    Dim i As Long
    Dim c As Collection
    Dim d As Collection
    
    For i = 1 To 100000
        Set c = New Collection
        Set d = New Collection
        
        c.Add d
        d.Add c
        '↑循環参照
        
        Debug.Print ObjPtr(c)
        Debug.Print ObjPtr(d)
        '↑それぞれのメモリアドレスが返る
    Next
    
'プロシージャが終了するタイミングでも、対象のオブジェクトは破棄されずメモリ解放されない
End Sub
```
```
Sub loopTest3()
    Dim i As Long
    Dim c As Collection
    Dim d As Collection
    
    For i = 1 To 100000
        Set c = New Collection
        Set d = New Collection
        
        c.Add d
        d.Add c
        '↑循環参照
        
        Debug.Print ObjPtr(c)
        Debug.Print ObjPtr(d)
        '↑それぞれのメモリアドレスが返る
    Next
    
    End 'Endステートメントを呼び出しても対象のオブジェクトは破棄されずメモリ解放されない
End Sub
```
※上記2つのコードの実行結果は先ほどと同様なため省略します。


これはオブジェクト変数から対象のオブジェクトへの参照が無くなるのであって、オブジェクトそのもの同士に持たせた参照は無くならないためです。
そのため、循環参照がある場合はそれを解消する必要があります。
循環参照を解消したのち、対象のオブジェクト変数へNothingを代入し、オブジェクトへの参照を無くすことで、オブジェクトを破棄しメモリ解放すると良いです。
```
Sub loopTest4()
    Dim i As Long
    Dim c As Collection
    Dim d As Collection
    
    For i = 1 To 100000
        Set c = New Collection
        Set d = New Collection
        
        c.Add d
        d.Add c
        '↑循環参照
        
        Debug.Print ObjPtr(c)
        Debug.Print ObjPtr(d)
        '↑それぞれのメモリアドレスが返る
        
        c.Remove 1 'cからdをRemoveすることで循環参照を解消する
        
        Debug.Print ObjPtr(c)
        Debug.Print ObjPtr(d)
        '↑それぞれ先ほどと同じメモリアドレスが返る
        
        Set d = Nothing
        Set c = Nothing
        
        Debug.Print ObjPtr(c) '0
        Debug.Print ObjPtr(d) '0
    Next
End Sub
```
上記コードの実行前後でメモリ使用量を比較してみると、実行前後でメモリ使用量はあまり変わらないことが確認できます。
- 実行前
![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/768789/de7ddc4a-89ba-f6a4-4794-c4669bfa2e5b.png)
- 実行後
![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/768789/f1940a74-ba38-f325-6b5b-0253efb080c8.png)


### 7-2. 外部参照のオブジェクトがある場合について
IE等の外部参照のオブジェクトは、オブジェクト変数からオブジェクトへの参照は無くなっても、オブジェクトは破棄されずメモリ解放されない場合があります。

例えばVBAからIEを起動し、プロシージャが終了してオブジェクト変数からオブジェクト(IE)への参照は無くなっても、IEのアプリケーション自体(=オブジェクト)は閉じられません。
つまり、オブジェクトは破棄されません。
```
Sub OpenIE()
    Dim IE As Object
    Set IE = CreateObject("InternetExplorer.Application")
    
    Debug.Print ObjPtr(IE)
    Debug.Print TypeName(IE)
    
    With IE
        .Visible = True 'TrueにしてIEを表示
        .navigate "https://qiita.com/" '任意のURLに変更
    End With
End Sub 'IEは閉じられない
```
この際、オブジェクト変数にNothingを代入したり、Endステートメントを呼び出しても結果は変わりません。
```
Sub OpenIE()
    Dim IE As Object
    Set IE = CreateObject("InternetExplorer.Application")
    
    Debug.Print ObjPtr(IE)
    Debug.Print TypeName(IE) 'オブジェクト名はIWebBrowser2
    
    With IE
        .Visible = True 'TrueにしてIEを表示
        .navigate "https://qiita.com/" '任意のURLに変更
    End With
    
    Set IE = Nothing
    Debug.Print ObjPtr(IE) '0
    Debug.Print TypeName(IE) 'Nothing
End Sub 'Nothingを代入してもIEは閉じられない
```
```
Sub OpenIE()
    Dim IE As Object
    Set IE = CreateObject("InternetExplorer.Application")
    
    Debug.Print ObjPtr(IE)
    Debug.Print TypeName(IE) 'オブジェクト名はIWebBrowser2
    
    With IE
        .Visible = True 'TrueにしてIEを表示
        .navigate "https://qiita.com/" '任意のURLに変更
    End With
    
    End 'Endステートメントを呼び出してもIEは閉じられない
End Sub
```

そのため、IE等の外部参照のオブジェクトを破棄するには、その外部参照のオブジェクト自身が持つクローズメソッドを実行する必要があります。
IEの場合では、Quitメソッドの実行が必要となります。

なお、クローズメソッドを実行してもオブジェクト変数からオブジェクトへの参照は残っているので、クローズメソッドを実行するだけではオブジェクトは破棄されません。
クローズメソッドを実行したうえで、プロシージャが終了するか、オブジェクト変数にNothingを代入し、オブジェクトへの参照がなくなることで、オブジェクトは破棄されメモリ解放されます。
```
Sub OpenIE()
    Dim IE As Object
    Set IE = CreateObject("InternetExplorer.Application")
    
    Debug.Print ObjPtr(IE)
    Debug.Print TypeName(IE) 'オブジェクト名はIWebBrowser2
    
    With IE
        .Visible = True 'TrueにしてIEを表示
        .navigate "https://qiita.com/" '任意のURLに変更
    End With
    
    IE.Quit 'IEを閉じる
    Debug.Print ObjPtr(IE) 'IEを閉じてもオブジェクトへの参照は残っている
    Debug.Print TypeName(IE) 'オブジェクト名はObject
    
    Set IE = Nothing
    Debug.Print ObjPtr(IE) '0
    Debug.Print TypeName(IE) 'Nothing
End Sub
```


### 7-3. Workbookがある場合について
#### 7-3-1. オブジェクトの親子関係について
まずオブジェクトの親子関係について説明します。
オブジェクトは以下のように親子関係(階層構造)になっています。
Application
└Workbook
　└Worksheet
　　└Range

オブジェクトの親子関係はオブジェクトそのものに存在しています。
親/子オブジェクトのオブジェクト変数には、それらへの参照が入っているだけであり、どちらへ先にNothingを代入しても最終的にそれらのオブジェクト変数からオブジェクトへの参照は無くなります。
そのため、Nothingを代入する順番は気にする必要はありません。

オブジェクトの親子関係はオブジェクトそのものに存在しているかを実際に確認してみます。

Workbook/WorksheetオブジェクトそれぞれをWB/WS変数へ代入した場合で考えます。
この場合、WorkbookオブジェクトとWB変数、WorksheetオブジェクトとWS変数のメモリアドレスは同じです。
そのため、Workbookオブジェクトを使用してWS変数を作成するのと、WB変数を使用してWS変数を作成するのは同義となります。
![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/768789/8ba5471a-908c-f593-e864-7cba3632dc21.png)
![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/768789/1e176b75-c1d0-2656-9efc-13733ff18174.png)

WB/WS変数の親オブジェクト名をParentプロパティで確認してみると、WB変数は**Application**、WS変数は**Workbook**が返ります。
これはWB/WS変数それぞれの参照先オブジェクトの親オブジェクト名が返っています。

変数宣言直後のWB/WS変数のオブジェクト名を確認してみると、それぞれ**Nothing**となっています。
ObjPtr関数でWB/WS変数のメモリアドレスを確認すると0が返ります。
これはまだ何も参照していないためです。

また、Workbookそのものを閉じてからWB/WS変数のオブジェクト名を確認してみると、それぞれ**Object**となっています。[^Object]
ObjPtr関数でWB/WS変数のメモリアドレスを確認するとWorkbookを閉じる前後で同一であることが確認できます。
つまり、Workbookそのものを閉じてもWB/WS変数からオブジェクトへの参照は残っています。
![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/768789/f6084011-f106-142f-840b-25b70c5c62b0.png)

変数宣言直後とWorkbookそのものを閉じた後にWB/WS変数へParentプロパティを実行すると、どちらの場合でもエラー発生するため、**Nothing**や**Object**には親子関係は存在しないことが確認できます。
そのため、オブジェクトの親子関係はオブジェクトそのものに存在しているということが確認できます。
```
Sub wbTest()
    Dim wb As Workbook
    Dim ws As Worksheet
    
    Debug.Print TypeName(wb.Parent) 'エラー発生
    Debug.Print TypeName(ws.Parent) 'エラー発生
    
    Set wb = Workbooks.Open(Filename:="C:\test\test.xlsx")
    Set ws = wb.Sheets("Sheet1")
    
    wb.Close
    
    Debug.Print TypeName(wb.Parent) 'オートメーションエラー発生
    Debug.Print TypeName(ws.Parent) 'オートメーションエラー発生
End Sub
```

#### 7-3-2. Workbook/Worksheetオブジェクトの場合について
Workbookそのものを閉じてもWB/WS変数からオブジェクトへの参照は残っているということは、Workbookを閉じるだけではオブジェクトは破棄されません。
プロシージャが終了するか、WB/WS変数にNothingを代入し、オブジェクトへの参照がなくなることで、オブジェクトは破棄されメモリ解放されます。[^Workbook/Worksheetの場合]

また、Workbookを閉じずにプロシージャが終了しても、Workbookを閉じずにWB/WS変数へNothingを代入しても、どちらの場合でもWorkbookそのものが残っているため、オブジェクトは破棄されません。これはEndステートメントを呼び出した場合でも同様です。
つまり、Workbook/Worksheetオブジェクトの場合、オブジェクトへの参照がなくなり、Workbookそのものが閉じられることで、オブジェクトが破棄されメモリ解放されます。[^順不同]　[^Range]

ちなみに、Workbookを一度開いてその後開きっぱなしの場合、プロシージャを何度実行してもWorkbook/Worksheetのメモリアドレスは毎回同じです。[^Workbook/Worksheet]
![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/768789/3c0c0ec1-2f37-954c-8789-6d7796830562.png)

そのため**WB/WS変数①**にNothingを代入後、**WB/WS変数②**にWorkbook/Worksheetオブジェクトを代入してみると、WB/WS変数①と②のメモリアドレスはどちらも同じであることが確認できます。
![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/768789/97960394-0502-6356-31cc-e95a546e3f57.png)


# 8. おわりに
このように、**Set オブジェクト変数=Nothing**は必要かという話は、状況によるうえ複雑であり、全て理解したうえで都度判断するというのはなかなか大変だと思います。
そのため理解しているに越したことはありませんが、冒頭で書いた結論の通りルール化すると良いと思います。


# 9. おまけ
以下は関連する内容ではありますが、本記事の趣旨から少しずれるため、おまけとして記載します。

### 9-1. オブジェクト変数にオブジェクト変数を代入した場合について
オブジェクト変数にオブジェクト変数を代入した場合、オブジェクトのメモリアドレスのコピーとなります。
そのため、代入元/代入先のオブジェクト変数の参照先のオブジェクトは結果的に同じものとなります。

また、オブジェクト変数にオブジェクト変数を代入した後に、片方のオブジェクト変数にNothingを代入しても、Nothingを代入したオブジェクト変数からオブジェクトへの参照が無くなるだけで、もう片方のオブジェクト変数からオブジェクトへの参照は残ったままとなります。
そのため対象のオブジェクトを破棄したい場合、もう片方のオブジェクト変数にもNothingを代入する必要があります。
なお、この場合でもプロシージャが終了するタイミングでそれらのオブジェクト変数からオブジェクトへの参照は無くなります。
![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/768789/dfd194ca-f256-17c6-e5c7-e7fd831c5e1f.png)


### 9-2. Withステートメントでのオブジェクトの扱いについて
Withステートメントは、指定したオブジェクトへの参照をWithブロック内で保持します。
例えば、オブジェクト変数を作成し、そのオブジェクト変数をWithステートメントで指定したとします。
その後Withブロック内でそのオブジェクト変数へNothingを代入すると、オブジェクト変数からオブジェクトへの参照は無くなりますが、Withブロック内でオブジェクトへの参照は続いているため、Withブロックを抜けるまでWithステートメントからオブジェクトへの参照は無くなりません。
![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/768789/8559d2bf-aad7-4e13-22f8-bcd4e91537fc.png)
![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/768789/a962b28b-f813-b6fa-742f-c3c688c8b33b.png)

ちなみにWithステートメントで保持するのはオブジェクトへの参照(メモリアドレス)なので、オブジェクトとオブジェクト変数のどちらを指定しても同じです。
![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/768789/2d4aa7f8-5051-f9a6-ab7d-dc3ad5d2e9fa.png)


# 10. 参考記事

https://thom.hateblo.jp/entry/2015/12/20/135035

https://thom.hateblo.jp/entry/2016/10/06/003147

https://thom.hateblo.jp/entry/2016/10/06/202616

https://www.relief.jp/docs/excel-vba-what-is-nothing.html

https://www.moug.net/tech/exvba/0150027.html

https://darumaexcel.uijin.com/aboutooa/object_delete.html

https://www.kurumico.com/excel-vba-class11/12406/#toc4

https://qiita.com/MitAtoM/items/36c3672248fb50554c9d

https://excel-ubara.com/excelvba4/EXCEL_VBA_422.html

https://www.relief.jp/docs/excel-vba-parent-property.html

https://learn.microsoft.com/ja-jp/office/vba/language/reference/user-interface-help/nothing-keyword

https://www.vba-ie.net/statement/end.php#google_vignette

https://qiita.com/pseudo_foxkeh/items/a15e4a5fd6ba679d4841

https://qiita.com/nekohei/items/0bb48a1bb60e2b198efc


[^メモリアドレス]:ここでいうメモリアドレスとは、物理メモリアドレスではなく、OSによって管理されたプロセスごとの仮想メモリ空間のアドレスです。
[^メモリ領域]:オブジェクトはメモリ内のヒープ領域に保存されるようです。ちなみに、ローカル変数はメモリ内のスタック領域を使用しているようです。
[^GC]:私は**ガーベジコレクション**派です(英語で**Garbage Collection**のため)。
[^Endステートメント]:**End Sub**や**End Function**などとは別物です。
それらはSubやFunctionといったプロシージャレベルのものを終了させますが、**End**はプログラムそのものを強制終了させます。
[^Object]:この**Object**は、Workbook/Worksheetオブジェクトのために確保されたメモリ領域が、Workbookそのものを閉じた後もオブジェクト変数から参照され続けていることで、保持されているものと考えられます。
[^Workbook/Worksheet]:Workbook/WorksheetオブジェクトはWorkbook/Worksheetそのものなため。
[^Workbook/Worksheetの場合]:外部参照のオブジェクトがある場合と似ています。
[^順不同]:Workbook/Worksheetオブジェクトへの参照がなくなるのと、Workbookそのものを閉じるのはどちらが先でも問題ありません。
[^Range]:今回記載していませんが、Rangeオブジェクトの場合もWorkbook/Worksheetオブジェクトの場合と同様です。
[^再代入]:念のため、オブジェクト変数へオブジェクトを再代入する前にも、オブジェクト変数へNothingを代入することを推奨します。